

PWM 波形のフーリエ解析を行う方法

ここでは、Visual Basic 言語によって記述した Excel のマクロのコードを用いて、インバータの PWM 波形（単相の場合、図 12.5、三相の場合、図 12.8）とその高調波解析（単相の場合、図 12.6、三相の場合、図 12.9）を計算、表示する方法を示す。

1. Excel を起動し、新規ブックを作成する
2. 表示→マクロ→マクロの表示→マクロ名を記載（たとえば「単相 PWM」）→作成
3. Microsoft Visual Basic for Applications が開くので、「Book1-Moduel1（コード）」画面の Sub と End Sub の間に、次ページの「マクロコード」を貼り付ける。

Sub 単相 PWM()

（ここに「マクロコード」を貼り付ける）

End Sub

単相か三相かの選択は「マクロコード」の以下 2 行のうち、いずれかをコメント（最初の文字に「'」を入力）にすればよい。

Cells(i + 2, 5) = -Cells(i + 2, 4) '単相

Cells(i + 2, 5) = amp * Sin(2 * pi * freq * i * dt - 2 * pi / 3) '三相

4. 実行→Sub/ユーザホームの実行→（最初 1 回目はエラーが出る）終了
5. Excel のブックに戻り、セル A1 の説明のように、セル A2～A6 の変数の値をセル B2～B6 に入力する（たとえば、セル A2 に記載された「基本波 f[Hz] 例 50」の「50」をセル B2 に入力する）。各セルの説明は以下のとおり。

- セル(B2)：変調波 vref の周波数 f [Hz]
- セル(B3)：搬送波 vc の周波数（変調波周波数 f の倍数で指定）
- セル(B4)：変調波 vref の振幅（0～1）。ただし vc の振幅を 1 とする
- セル(B5)：出力電圧 v のパルスの周期の分割数（約 100）。出力電圧 v のパルスはこの点数で近似してフーリエ解析する
- セル(B6)：フーリエ解析個数（変調波周波数 f の倍数で指定）
- セル(B6)：フーリエ表示個数（変調波周波数 f の倍数で指定）

6. Microsoft Visual Basic for Applications-Book1 にて実行→Sub/ユーザホームの実行
7. Excel のブックに、「搬送波、三角波、PWM 波」「フーリエ解析」「PWM 波、フーリエ解析から合成した波形」が表示される
8. 5, 6 を繰り返すことで、検討できる

マクロコード

```
' 正弦波波形, PWM 波形の作成
Dim freq, dt, amp, pi As Double
Dim n, n1, n2, n11 As Integer
Dim i, j, k As Integer
Range(Columns(3), Columns(14)).Clear
Range(Cells(10, 2), Cells(11, 2)).Clear
ActiveSheet.DrawingObjects.Delete
Cells(1, 1) = "6 パラメータを B2, . . . ,B7 に設定してください"
Cells(2, 1) = " 変調波周波数 f [Hz]          例 50"
Cells(3, 1) = " 搬送波周波数 (f の倍数で指定) 例 12"
Cells(4, 1) = " 変調波振幅 (0~1)      例 0.75"
Cells(5, 1) = " 1 パルスの分割数      例 100"
Cells(6, 1) = " フーリエ解析個数      例 200"
Cells(7, 1) = " フーリエ表示個数      例 80"

Cells(1, 3) = "t"
Cells(1, 4) = "sin  $\theta$  "
Cells(1, 5) = "- sin  $\theta$  "
Cells(1, 6) = "三角波"
Cells(1, 7) = "Va"
Cells(1, 8) = "Vb"
Cells(1, 9) = "Vab"
Cells(1, 10) = "Freq"
Cells(1, 11) = "Fourier of Vab"
Cells(1, 12) = "t"
Cells(1, 13) = "Vab"
Cells(1, 14) = "Vab by Fourier"

freq = Cells(2, 2).Value
n = Cells(3, 2).Value
amp = Cells(4, 2).Value
n1 = Cells(5, 2).Value '1 パルスを n1 分割する
pi = 3.14159265358979
Cells(10, 2).Value = 1 / freq
```

```

dt = 1 / (2 * n * freq * n1)
Cells(11, 2).Value = dt
n11 = 2 * n * n1
' 時間軸の作成
For i = 0 To n11
    Cells(i + 2, 3) = i * dt
Next i
' sin 波,-sin 波
For i = 0 To n11
    Cells(i + 2, 4) = amp * Sin(2 * pi * freq * i * dt)
    Cells(i + 2, 5) = -Cells(i + 2, 4) '単相
    Cells(i + 2, 5) = amp * Sin(2 * pi * freq * i * dt - 2 * pi / 3) '三相
Next i
' 三角波
For i = 0 To n - 1
    For j = 0 To n1 / 2
        k = i * 2 * n1 + j
        Cells(k + 2, 6) = j / (n1 / 2)
    Next j
    For j = 0 To n1
        k = i * 2 * n1 + n1 / 2 + j
        Cells(k + 2, 6) = 1 - j / (n1 / 2)
    Next j
    For j = 0 To n1 / 2
        k = i * 2 * n1 + n1 * 3 / 2 + j
        Cells(k + 2, 6) = -1 + j / (n1 / 2)
    Next j
Next i
' Va
For i = 0 To n11
    If Cells(i + 2, 4) >= Cells(i + 2, 6) Then
        k = 1
    Else
        k = 0
    End If
    Cells(i + 2, 7) = k

```

```

    If Cells(i + 2, 5) >= Cells(i + 2, 6) Then
        k = 1
    Else
        k = 0
    End If
    Cells(i + 2, 8) = k
    Cells(i + 2, 9) = Cells(i + 2, 7).Value - Cells(i + 2, 8)
Next i
' グラフ作成
With ActiveSheet.Shapes.AddChart.Chart
    .ChartType = xlXYScatterLinesNoMarkers
    .SetSourceData Range(Cells(1, 3), Cells(n11, 9))
End With

' フーリエ解析
Dim t(5000), f(5000), aa(5000), bb(5000), four(1000) As Double
Dim istep, istep1, nn, nn2 As Integer
Dim tall, f_ave, an, bn, a, b, c, d, a0, a1, b0, b1, q_a, ff, x As Double
istep = n11
nn = Cells(6, 2).Value
nn2 = Cells(7, 2).Value
istep1 = istep + 1

' 時間データ入力
For i = 1 To istep1
    t(i) = Cells(i + 1, 3)
    f(i) = Cells(i + 1, 9)
Next i
tall = t(istep1) - t(1)

' 平均値
f_ave = 0
For i = 1 To istep
    f_ave = f_ave + f(i)
Next i
f_ave = f_ave / istep
Cells(2, 10) = 0

```

```

Cells(2, 11) = f_ave
' フーリエ計算
For n = 1 To nn
    an = 0
    bn = 0
    For i = 1 To istep
        a = (f(i + 1) - f(i)) / (t(i + 1) - t(i))
        b = -a * t(i) + f(i)
        c = n * pi * 2 / (t(istep + 1) - t(1))
        d = -n * pi * (t(istep + 1) + t(1)) / (t(istep + 1) - t(1))
        a0 = (a / c ^ 2 * Cos(c * t(i + 1) + d) + a * t(i + 1) / c * Sin(c * t(i + 1) + d)) -
(a / c ^ 2 * Cos(c * t(i) + d) + a * t(i) / c * Sin(c * t(i) + d))
        a1 = (b / c * Sin(c * t(i + 1) + d)) - (b / c * Sin(c * t(i) + d))
        b0 = (a / c ^ 2 * Sin(c * t(i + 1) + d) - a * t(i + 1) / c * Cos(c * t(i + 1) + d)) -
(a / c ^ 2 * Sin(c * t(i) + d) - a * t(i) / c * Cos(c * t(i) + d))
        b1 = (-b / c * Cos(c * t(i + 1) + d)) - (-b / c * Cos(c * t(i) + d))
        an = an + 2 / (t(istep + 1) - t(1)) * (a0 + a1)
        bn = bn + 2 / (t(istep + 1) - t(1)) * (b0 + b1)
    Next i
    four(n) = Sqr(an ^ 2 + bn ^ 2)
    aa(n) = an
    bb(n) = bn
    ff = n / tall
    Cells(2 + n, 10) = ff
    Cells(2 + n, 11) = four(n)
Next n
' グラフ作成
With ActiveSheet.Shapes.AddChart.Chart
    .ChartType = xlXYScatterLinesNoMarkers
    .SetSourceData Range(Cells(1, 10), Cells(nn2 + 2, 11))
End With

' フーリエ解析から元の波形を合成してチェック
For i = 0 To istep1
    x = i * 2 * pi / istep1 - pi
    a0 = 0

```

```

For n = 1 To nn
    a0 = a0 + aa(n) * Cos(n * x) + bb(n) * Sin(n * x)
Next n
'Cells(2 + i, 12) = cells(2+i,3)
Cells(2 + i, 12) = t(1) + (x + pi) * 0.5 * tall / pi
Cells(2 + i, 13) = Cells(2 + i, 9)
Cells(2 + i, 14) = a0 + f_ave
Next i
' グラフ作成
With ActiveSheet.Shapes.AddChart.Chart
    .ChartType = xlXYScatterLinesNoMarkers
    .SetSourceData Range(Cells(1, 12), Cells(n11, 14))
End With

With ActiveSheet
    .ChartObjects(1).Top = Range("c3").Top
    .ChartObjects(1).Left = Range("c3").Left '位置を設定
    .ChartObjects(2).Top = Range("e6").Top
    .ChartObjects(2).Left = Range("e6").Left '位置を設定
    .ChartObjects(3).Top = Range("g9").Top
    .ChartObjects(3).Left = Range("g9").Left '位置を設定
End With

```